Example 8b: Three-Level Models for Intensive Longitudinal Data in SAS

These are real data borrowed from the Affect Health and Behavior Study (AHAB) conducted by KU PI Chris Cushing's Pediatric Health Insight Team (PHIT). The present sample consists of 25 adolescents who completed surveys about their mood and energy levels up to four times per day for up to 20 days (total *N* = 976). In this example we will examine how self-reported fatigue varies throughout the day, how negative affect relates to fatigue, and how variation in sleep may moderate these relationships. Given our use of last night's sleep as a predictor and concerns about first-day reactivity, only days 3–20 are used here. The variables are as follows:

- *Hours of Sleep*: previous night's sleep duration (day-level variable) → moderator
- *Fatigue*: measured four times per day using 3 items (each response from 1–5) → outcome
- *Negative Affect*: measured four times per day using 5 items (each response from 1–5) → predictor
- *SinceMidnight*: time of observation in hours since midnight → time predictor

Code for Data Manipulation:

```
* Build ID-day and time-related variables;
DATA work.EMA; SET work.EMA; FORMAT Date MMDDYY8.; * Format date;
       PersonIDday = PersonID + Day/100;
       SinceMidnight = TimeOfDay/3600;
       DayOfWeek = WEEKDAY(Date); * 1=Sunday through 7=Saturday;
       IF DayOfWeek=2 THEN Monday=1; ELSE Monday=0;
* Sort by person, day, time;
PROC SORT DATA=work.EMA; BY PersonIDday TimeOfDay; RUN;

* Get means per day;
PROC MEANS NOPRINT DATA=work.EMA; BY PersonID PersonIDday Day;
       VAR Day HoursSleep NegAffect SinceMidnight; OUTPUT OUT=work.DayMeans N(Day)=Nperday
              MEAN(HoursSleep NegAffect SinceMidnight)= DaySleep DayNegAffect DaySinceMidnight; RUN;
* Create lagged day sleep;
PROC EXPAND DATA=work.DayMeans OUT=work.DayMeansLag;
       BY PersonID; ID Day;  * Old = new name;
       CONVERT DaySleep=LagDaySleep / METHOD=NONE TRANSFORMOUT=(LAG 1); RUN;
* Get person means;
PROC MEANS NOPRINT DATA=work.DayMeansLag; BY PersonID;
       VAR Day DaySleep DayNegAffect DaySinceMidnight; OUTPUT OUT=work.PersonMeans N(Day)=Ndays
       MEAN(DaySleep DayNegAffect DaySinceMidnight)=PersonSleep PersonNegAffect PersonSinceMidnight;
RUN;
* Merge person means into day means, create level-3 and level-2 variables;;
DATA work.DayMeans; MERGE work.PersonMeans work.DayMeansLag; BY PersonID; DROP _TYPE_ _FREQ_;
       L3Sleep=PersonSleep-7; L2Sleep=DaySleep-PersonSleep;
       L3NA=PersonNegAffect-10; L2NA=DayNegAffect-PersonNegAffect;
       L3T=PersonSinceMidnight-15; RUN;
* Merge day means into level-1 data and create level-variables;
DATA work.EMA; MERGE work.EMA work.DayMeans; BY PersonIDday;
       L1NA=NegAffect-DayNegAffect; L1T=SinceMidnight-15;
       RoundTimeofDay=ROUND(SinceMidnight,.5); RUN;

* Subset to days 2-20 and complete cases;
DATA work.EMA; SET work.EMA; IF Day=1 THEN DELETE;
       WHERE NMISS(Fatigue,Day,L1T,L1NA,L2NA,L3NA,L2Sleep,L3Sleep)=0; RUN;
* Ranges for variables;
PROC MEANS NONOBS NDEC=2 DATA=work.EMA; VAR Fatigue NegAffect DaySleepLag SinceMidnight; RUN;
```

Variable	N	Mean	Std Dev	Minimum	Maximum
Fatigue	976	6.11	3.21	3.00	15.00
NegAffect	976	7.51	4.19	5.00	25.00
LagDaySleep	976	6.80	1.35	3.10	11.45
SinceMidnight	976	14.85	4.93	6.20	23.56

Model 1a: Two-Level Empty Means, Random Intercept for the Level-1 Fatigue Outcome

Level 1 Time: $\text{Fatigue}_{tdi} = \beta_{0di} + e_{tdi}$
Level 2 Day: Intercept: $\beta_{0di} = \delta_{00i}$
Level 3 Person: Intercept: $\delta_{00i} = \gamma_{000} + V_{00i}$

```
TITLE1 "Model 1a: Empty Means, Two-Level Model for Fatigue Outcome";
TITLE2 "All Occasions at Level 1 within Persons at Level 2";
PROC MIXED DATA=work.EMA COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
       CLASS PersonID Day;
       MODEL Fatigue =  / SOLUTION DDFM=Satterthwaite;
       RANDOM INTERCEPT / VCORR TYPE=UN SUBJECT=PersonID;
       ODS OUTPUT InfoCrit=FitDV2level; * Save for fit; RUN;
```

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	PersonID	6.7127	2.0769	3.23	0.0006
Residual		6.1046	0.2799	21.81	<.0001

Calculate the ICC_{L3b} for the correlation of occasions from the same person:

$$ICC = \frac{6.7127}{6.7127 + 6.1046} = .5237$$

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
1	429.42	<.0001

This LRT tells us that the random intercept variance is significantly greater than 0, and thus so is the ICC.

Model 1b: Three-Level Empty Means, Random Intercept for the Level-1 Fatigue Outcome

Level 1 Time: $\text{Fatigue}_{tdi} = \beta_{0di} + e_{tdi}$
Level 2 Day: Intercept: $\beta_{0di} = \delta_{00i} + U_{0di}$
Level 3 Person: Intercept: $\delta_{00i} = \gamma_{000} + V_{00i}$

```
TITLE1 "Model 1b: Empty Means, Three-Level Model for Fatigue Outcome";
TITLE2 "Level-1 Occasions within Level-2 Days within Level-3 Persons";
PROC MIXED DATA=work.EMA COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
       CLASS PersonID Day;
       MODEL Fatigue =  / SOLUTION DDFM=Satterthwaite;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PersonID;       * Level 3;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PersonID*Day;   * Level 2;
       ODS OUTPUT InfoCrit=FitDV3level CovParms=CovEmpty; * Save for fit and pseudoR2; RUN;
* Test 2-level vs 3-level;
%FitTest(FitFewer=FitDV2level, FitMore=FitDV3level);
```

Is the 3-level model a better fit than the 2-level model?
Yes, $-2\Delta LL(\sim1) = 75.76$, $p < .001$, so keep it

Likelihood Ratio Test for FitDV2level vs. FitDV3level

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitDV2level	4620.1	2	4624.1	4626.5	.	.	.
FitDV3level	4544.3	3	4550.3	4553.9	75.7588	1	0

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	PersonID	6.4470	2.0665	3.12	0.0009 L3 int
UN(1,1)	PersonID*Day	1.6826	0.2763	6.09	<.0001 L2 int
Residual		4.5427	0.2409	18.86	<.0001 L1 res

Proportion variance at each level:
Total = 6.4470 + 1.6826 + 4.5427 = 12.672
Level 3 (person) = 6.4470 / 12.672 = .51
Level 2 (day) = 1.6826 / 12.672 = .13
Level 1 (time) = 4.5427 / 12.672 = .36

ICC_{L3b} for proportion of between-person variance over total variance
= 6.4470 / 12.672 = .51

ICC_{L2b} for proportion of between-day over within-person variance
= 1.6826 / (1.6826 + 4.5427) = .27
This ICC_{L2b} = .27 is significantly greater than 0 via $-2\Delta LL$ for 3- vs. 2-level.

Now let's do the same thing for our level-2 moderator of last night's sleep.
A Two-Level Model indicates we will need to represent sleep at levels 2 and 3.

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	PersonID	0.5421	0.2022	2.68	0.0037
Residual		1.4686	0.06754	21.74	<.0001

Calculate the ICC for the correlation of days from the same person:
$$ICC = \frac{0.5421}{0.5421 + 1.4686} = .2696$$

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
1	157.73	<.0001

This LRT tells us that the random intercept variance is significantly greater than 0, and thus so is the ICC.

Now let's do the same thing for our level-1 predictor of negative affect. Two-Level Model Results:

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	PersonID	19.5994	5.8817	3.33	0.0004
Residual		5.8647	0.2688	21.82	<.0001

Calculate the ICC_{L3b} for the correlation of occasions from the same person:
$$ICC = \frac{19.5994}{19.5994 + 5.8647} = .7697$$

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
1	960.55	<.0001

This LRT tells us that the random intercept variance is significantly greater than 0, and thus so is the ICC.

Three-Level Model Results indicate we will need to represent negative affect at levels 1, 2, and 3:

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	PersonID	19.3533	5.8388	3.31	0.0005 → 76.61%, so ICCL3b = .7661
UN(1,1)	PersonID*Day	0.7309	0.2028	3.60	0.0002 → 2.89%, so ICCL2b = .1237
Residual		5.1785	0.2720	19.04	<.0001 → 20.50%

Likelihood Ratio Test for FitIV2level vs. FitIV3level

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitIV2level	4605.9	2	4609.9	4612.3	.	.	.
FitIV3level	4587.6	3	4593.6	4597.1	18.3342	1	.000018535

Is the 3-level model a better fit than the 2-level model?
Yes, $-2\Delta LL(\sim 1) = 18.33$, $p < .001$, so keep it

Now let's do the same thing for our level-1 predictor of time of day. Two-Level Model Results:

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	PersonID	0.5801	0.3467	1.67	0.0471
Residual		23.7931	1.0883	21.86	<.0001

Calculate the ICC_{L3b} for the correlation of occasions from the same person:
$$ICC = \frac{0.5801}{0.5801 + 23.7931} = .0238$$

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
1	7.71	<.0001

This LRT tells us that the random intercept variance is significantly greater than 0, and thus so is the ICC.

Three-Level Model says time is needed at levels 1 and 3 only:

Is the 3-level model a better fit?
Not if it can't find any level-2, day variance!

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	PersonID	0.5798	0.3465	1.67	0.0471 → 2.38%, so ICCL3b = .0238
UN(1,1)	PersonID*Day	0	.	.	. → 0%, so ICCL2b = 0
Residual		23.7932	1.0883	21.86	<.0001 → 97.62%

Now we need to examine unconditional models for possible changes over "time" in our fatigue outcome: this means examining different types of change across level 1 and level 2. To do so, I am temporarily switching to GLIMMIX to get plots from LSMEANS (which are not available in my version of MIXED).

```
TITLE1 "Examine Saturated Means for Level-1 Within-Day Change";
PROC GLIMMIX DATA=work.EMA NOCLPRINT NAMELEN=100 METHOD=RSPL;
       CLASS PersonID Day RoundTimeofDay;
       MODEL Fatigue = RoundTimeofDay / SOLUTION DDFM=Satterthwaite;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PersonID;          * Level 3;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PersonID*Day;      * Level 2;
       LSMEANS RoundTimeofDay / PLOT=MEANPLOT(CLBAND JOIN);  RUN;
```



It looks like fatigue has a quadratic pattern of change over the day (i.e., it is highest in morning and evening). Plotting the means in excel and adding a polynomial trend line supports this idea.




```
TITLE1 "Examine Saturated Means for Level-2 Across-Day Change";
PROC GLIMMIX DATA=work.EMA NOCLPRINT NAMELEN=100 METHOD=RSPL;
       CLASS PersonID Day;
       MODEL Fatigue = Day / SOLUTION DDFM=Satterthwaite;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PersonID;          * Level 3;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PersonID*Day;      * Level 2;
       LSMEANS Day / PLOT=MEANPLOT(CLBAND JOIN);  RUN;
```



There doesn't appear to be any systematic trend by day of study. Plotting the means in excel and adding a polynomial trend line (which is flat) supports this idea.




```
TITLE1 "Examine Saturated Means for Level-2 Day of Week Change";
PROC GLIMMIX DATA=work.EMA NOCLPRINT NAMELEN=100 METHOD=RSPL;
       CLASS PersonID Day DayOfWeek;
       MODEL Fatigue = DayOfWeek / SOLUTION DDFM=Satterthwaite;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PersonID;          * Level 3;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PersonID*Day;      * Level 2;
       LSMEANS DayOfWeek / DIFF=ALL PLOT=MEANPLOT(CLBAND JOIN);  RUN;
```



Day 2 is slightly higher, so it looks like this sample might have a case of the Mondays! ☺

Based on the previous plots, I propose the following unconditional model for time as Model 2a:

Level 1 Time: $\text{Fatigue}_{tdi} = \beta_{0di} + \beta_{1di}(\text{Hour}_{tdi} - 15) + \beta_{2di}(\text{Hour}_{tdi} - 15)^2 + e_{tdi}$

Level 2 Day: Intercept: $\beta_{0di} = \delta_{00i} + \delta_{01i}(\text{Monday}_{di}) + U_{0di}$

Linear Time: $\beta_{1di} = \delta_{10i}$

Quadratic Time: $\beta_{2di} = \delta_{20i}$

Level 3 Person: Intercept: $\delta_{00i} = \gamma_{000} + \gamma_{001}(\overline{\text{Hour}}_i - 15) + V_{00i}$

Linear Time: $\delta_{10i} = \gamma_{100} + \gamma_{101}(\overline{\text{Hour}}_i - 15)$

Quadratic Time: $\delta_{20i} = \gamma_{200} + \gamma_{201}(\overline{\text{Hour}}_i - 15)$

It's Monday: $\delta_{01i} = \gamma_{010}$

The new fixed effects include a L2 dummy code for if it's Monday, as well as linear and quadratic L1 time of day (predictors labeled "T", where 0=3 PM). The linear L3 time effects (also 0=3PM) are needed to create contextual effects given the use of constant-based-centering for L1 time. Quadratic L3 time main and interaction effects were nonsignificant.

```
TITLE1 "Model 2a: Add Fixed Effects of L2 Monday, Quadratic Time of Day (0=3PM) at L1 and L3";
PROC MIXED DATA=work.EMA COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
       CLASS PersonID Day;
       MODEL Fatigue = Monday L1T L1T*L1T L3T L1T*L3T L1T*L1T*L3T
                        / SOLUTION DDFM=Satterthwaite OUTPM=PredTime; * Save for total-R2;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PersonID;         * Level 3;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PersonID*Day;     * Level 2;
       ODS OUTPUT InfoCrit=FitFixTime CovParms=CovFixTime; * Save for fit and pseudo-R2; RUN;
```

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	PersonID	6.1526	2.0326	3.03	0.0012 L3 intercept
UN(1,1)	PersonID*Day	1.6220	0.2677	6.06	<.0001 L2 intercept
Residual		4.4019	0.2340	18.81	<.0001 L1 residual

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	5.8232	0.5416	22.2	10.75	<.0001 → Note that DDF ~ L3 effect
Monday	0.6911	0.2908	255	2.38	0.0182 → Note that DDF ~ L2 effect
L1T	0.008223	0.01666	760	0.49	0.6218 → Note that DDF ~ L1 effect
L1T*L1T	0.01634	0.003880	772	4.21	<.0001 → Note that DDF ~ L2 effect
L3T	0.8829	0.4828	22.3	1.83	0.0809 → Note that DDF ~ L3 effect
L1T*L3T	-0.01382	0.01537	751	-0.90	0.3689 → Note that DDF ~ L1 effect
L1T*L1T*L3T	-0.00752	0.003581	754	-2.10	0.0360 → Note that DDF ~ L1 effect

```
* Pseudo-R2 for time effects;
%PseudoR2(NCov=3, CovFewer=CovEmpty, CovMore=CovFixTime);
* Total-R2 for time effects (no "fewer" comparison possible here);
%TotalR2(DV=Fatigue, PredFewer=PredTime, PredMore=PredTime);
```

Name	Pred Corr	TotalR2
PredTime	0.28697	0.082354

PsuedoR2 (% Reduction) for CovEmpty vs. CovFixTime

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovEmpty	UN(1,1)	PersonID	6.4470	2.0665	3.12	0.0009	.
CovEmpty	UN(1,1)	PersonID*Day	1.6826	0.2763	6.09	<.0001	.
CovEmpty	Residual		4.5427	0.2409	18.86	<.0001	.
CovFixTime	UN(1,1)	PersonID	6.1526	2.0326	3.03	0.0012	0.045655 for L3 int
CovFixTime	UN(1,1)	PersonID*Day	1.6220	0.2677	6.06	<.0001	0.036025 for L2 int
CovFixTime	Residual		4.4019	0.2340	18.81	<.0001	0.031005 for L1 res

---

Let's test if the effects of level-1 time of day vary randomly over days and persons. Given the heavily quadratic time-of-day trend shown above, I'm going to break my own rules by testing random quadratic effects first, and then test if random linear effects are also needed. Only lines of code that changed are shown.

$$\text{Level 1 Time:}\quad \text{Fatigue}_{tdi} = \beta_{0di} + \beta_{1di}(\text{Hour}_{tdi} - 15) + \beta_{2di}(\text{Hour}_{tdi} - 15)^2 + e_{tdi}$$

$$\text{Level 2 Day:}\quad \text{Intercept:}\quad \beta_{0di} = \delta_{00i} + \delta_{01i}(\text{Monday}_{di}) + U_{0di}$$

$$\text{Linear Time:}\quad \beta_{1di} = \delta_{10i} + U_{1di} \quad \leftarrow \textbf{2c}$$

$$\text{Quadratic Time:}\quad \beta_{2di} = \delta_{20i} + U_{2di} \quad \leftarrow \textbf{2b}$$

$$\text{Level 3 Person: Intercept:}\quad \delta_{00i} = \gamma_{000} + \gamma_{001}(\overline{\text{Hour}}_i - 15) + V_{00i}$$

$$\text{Linear Time:}\quad \delta_{10i} = \gamma_{100} + \gamma_{101}(\overline{\text{Hour}}_i - 15)$$

$$\text{Quadratic Time:}\quad \delta_{20i} = \gamma_{200} + \gamma_{201}(\overline{\text{Hour}}_i - 15)$$

$$\text{It's Monday:}\quad \delta_{01i} = \gamma_{010}$$

Models 2b and 2c added random quadratic and then random linear time of day effects at level 2 only: across days.

LRTs reveal both the level-2 random linear and random quadratic effects are needed, and their $r = -0.56$ (so is ok).

```
TITLE1 "Model 2b: Add Random Effect of Only L1 Quadratic Time of Day Across L2 Days";
       RANDOM INTERCEPT L1T*L1T / GCORR TYPE=UN SUBJECT=PersonID*Day; * Level 2;
       ODS OUTPUT InfoCrit=FitR2QTime; * Save for fit; RUN;
* Test only random quadratic time across days; %FitTest(FitFewer=FitFixTime, FitMore=FitR2QTime);
```

Does the quadratic level-1 time effect vary randomly over level-2 days? *Yes, −2ΔLL(~2) =24.41, p < .001, so keep it*

Likelihood Ratio Test for FitFixTime vs. FitR2QTime

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitFixTime	4539.1	3	4545.1	4548.6	.	.	.
FitR2QTime	4514.7	5	4524.7	4530.6	24.4076	2	.000005011

```
TITLE1 "Model 2c: Add Random Effect of Also L1 Linear Time of Day Across L2 Days";
       RANDOM INTERCEPT L1T L1T*L1T / GCORR TYPE=UN SUBJECT=PersonID*Day; * Level 2;
       ODS OUTPUT InfoCrit=FitR2LQTime; * Save for fit; RUN;
* Test also random linear time across days; %FitTest(FitFewer=FitR2QTime, FitMore=FitR2LQTime);
```

Does the linear level-1 time effect vary randomly over level-2 days? *Yes, −2ΔLL(~3) =13.00, p = .005, so keep it*

Likelihood Ratio Test for FitR2QTime vs. FitR2LQTime

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitR2QTime	4514.7	5	4524.7	4530.6	.	.	.
FitR2LQTime	4501.7	8	4517.7	4527.1	12.9955	3	.004646336

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z	
UN(1,1)	PersonID	6.0888	2.0114	3.03	0.0012	L3 intercept
UN(1,1)	PersonID*Day	2.5582	0.5010	5.11	<.0001	L2 intercept
UN(2,1)	PersonID*Day	0.05112	0.04056	1.26	0.2075	
UN(2,2)	PersonID*Day	0.01140	0.006393	1.78	0.0373	L2 linear time
UN(3,1)	PersonID*Day	-0.02823	0.01258	-2.24	0.0248	
UN(3,2)	PersonID*Day	-0.00191	0.000979	-1.95	0.0514	
UN(3,3)	PersonID*Day	0.001012	0.000381	2.66	0.0039	L2 quad time
Residual		3.2864	0.2609	12.60	<.0001	L1 residual

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	
Intercept	5.8210	0.5396	22.4	10.79	<.0001	
Monday	0.6512	0.2854	241	2.28	0.0234	
L1T	0.006262	0.01627	234	0.38	0.7006	→ Note that DDF now ~ L2 effect
L1T*L1T	0.01560	0.004067	214	3.84	0.0002	→ Note that DDF now ~ L2 effect
L3T	0.9136	0.4812	22.5	1.90	0.0705	
L1T*L3T	-0.01286	0.01516	231	-0.85	0.3970	→ Note that DDF now ~ L2 effect
L1T*L1T*L3T	-0.00880	0.003788	197	-2.32	0.0212	→ Note that DDF now ~ L2 effect

---

Level 1 Time: $\text{Fatigue}_{tdi} = \beta_{0di} + \beta_{1di}(\text{Hour}_{tdi} - 15) + \beta_{2di}(\text{Hour}_{tdi} - 15)^2 + e_{tdi}$

Level 2 Day: Intercept: $\beta_{0di} = \delta_{00i} + \delta_{01i}(\text{Monday}_{di}) + U_{0di}$

Linear Time: $\beta_{1di} = \delta_{10i} + U_{1di}$

Quadratic Time: $\beta_{2di} = \delta_{20i} + U_{2di}$

Level 3 Person: Intercept: $\delta_{00i} = \gamma_{000} + \gamma_{001}(\overline{\text{Hour}}_i - 15) + V_{00i}$

Linear Time: $\delta_{10i} = \gamma_{100} + \gamma_{101}(\overline{\text{Hour}}_i - 15) + V_{10i}$ ← **2e**

Quadratic Time: $\delta_{20i} = \gamma_{200} + \gamma_{201}(\overline{\text{Hour}}_i - 15) + V_{20i}$ ← **2d**

It's Monday: $\delta_{01i} = \gamma_{010}$

Models 2d and 2e added random quadratic and then random linear time of day effects at level 3: across persons.

LRTs revealed only the level-3 random quadratic effect is needed. When I tried keeping the linear anyway, later on their *r* = −1 (so it became not ok).

```
TITLE1 "Model 2d: Add Random Effect of Only L1 Quadratic Time of Day Across L3 Persons";
       RANDOM INTERCEPT L1T*L1T / GCORR TYPE=UN SUBJECT=PersonID;  * Level 3;
       ODS OUTPUT InfoCrit=FitR3QTime CovParms=CovRandTime; * Save for fit and pseudo-R2; RUN;
* Test random quadratic time across persons; %FitTest(FitFewer=FitR2LQTime, FitMore=FitR3QTime);
```

Likelihood Ratio Test for FitR2LQTime vs. FitR3QTime

Does the quadratic level-1 time effect vary randomly over level-3 persons? *Yes, −2ΔLL(~2) =7.78, p = .020, so keep it*

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitR2LQTime	4501.7	8	4517.7	4527.1	.	.	.
FitR3QTime	4493.9	10	4513.9	4525.7	7.78333	2	0.020411

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z	
UN(1,1)	PersonID	6.5735	2.3371	2.81	0.0025	L3 intercept
UN(2,1)	PersonID	-0.01880	0.02424	-0.78	0.4381	
UN(2,2)	PersonID	0.000531	0.000331	1.60	0.0546	L3 quad time
UN(1,1)	PersonID*Day	2.4547	0.4782	5.13	<.0001	L2 intercept
UN(2,1)	PersonID*Day	0.03406	0.04112	0.83	0.4075	
UN(2,2)	PersonID*Day	0.01155	0.006327	1.82	0.0340	L2 linear time
UN(3,1)	PersonID*Day	-0.02225	0.01178	-1.89	0.0588	
UN(3,2)	PersonID*Day	-0.00137	0.000993	-1.38	0.1681	
UN(3,3)	PersonID*Day	0.000802	0.000354	2.27	0.0117	L2 quad time
Residual		3.2275	0.2550	12.66	<.0001	L1 residual

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	
Intercept	5.9301	0.5598	19.1	10.59	<.0001	
Monday	0.6352	0.2875	242	2.21	0.0281	
L1T	0.007288	0.01642	247	0.44	0.6576	→ Note that DDF still ~ L2 effect
L1T*L1T	0.01004	0.006471	14.6	1.55	0.1421	→ Note that DDF NOW ~ L3 effect
L3T	0.9528	0.4998	19.2	1.91	0.0716	
L1T*L3T	-0.01106	0.01540	246	-0.72	0.4732	→ Note that DDF still ~ L2 effect
L1T*L1T*L3T	-0.01064	0.005853	15.8	-1.82	0.0880	→ Note that DDF NOW ~ L3 effect

```
TITLE1 "Model 2e: Add Random Effect of Also L1 Linear Time of Day Across L3 Persons";
       RANDOM INTERCEPT L1T L1T*L1T / GCORR TYPE=UN SUBJECT=PersonID; * Level 3;
       ODS OUTPUT InfoCrit=FitR3LQTime; * Save for fit; RUN;
* Test also random linear time across persons; %FitTest(FitFewer=FitR3QTime, FitMore=FitR3LQTime);
```

Does the linear level-1 time effect vary randomly over level-3 persons? *No, $-2\Delta LL(\sim3)$ =2.31, p = .511, so drop it*

```
Likelihood Ratio Test for FitR3QTime vs. FitR3LQTime
               Neg2Log
   Name         Like      Parms        AIC        BIC     DevDiff     DFdiff      Pvalue
FitR3QTime     4493.9      10       4513.9     4525.7        .           .           .
FitR3LQTime    4491.6      13       4517.6     4532.9     2.30729        3        0.51113
```

$$\text{Level 1 Time:}\quad \text{Fatigue}_{tdi} = \beta_{0di} + \beta_{1di}(\text{Hour}_{tdi} - 15) + \beta_{2di}(\text{Hour}_{tdi} - 15)^2 + e_{tdi}$$
$$\text{Level 2 Day:}\quad \text{Intercept: } \beta_{0di} = \delta_{00i} + \delta_{01i}(\text{Monday}_{di}) + U_{0di}$$
$$\text{Linear Time: } \beta_{1di} = \delta_{10i} + U_{1di}$$
$$\text{Quadratic Time: } \beta_{2di} = \delta_{20i} + U_{2di}$$
$$\text{Level 3 Person: Intercept: } \delta_{00i} = \gamma_{000} + \gamma_{001}(\overline{\text{Hour}}_i - 15) + V_{00i}$$
$$\text{Linear Time: } \delta_{10i} = \gamma_{100} + \gamma_{101}(\overline{\text{Hour}}_i - 15)$$
$$\text{Quadratic Time: } \delta_{20i} = \gamma_{200} + \gamma_{201}(\overline{\text{Hour}}_i - 15) + V_{20i}$$
$$\text{It's Monday: } \delta_{01i} = \gamma_{010} + V_{01i}$$

Model 2f added a random Monday effect to level 3: across persons.

```
TITLE1 "Model 2f: Add Random Effect of L2 Monday Across L1 Persons";
       RANDOM INTERCEPT L1T*L1T Monday / GCORR TYPE=UN SUBJECT=PersonID; * Level 3;
       ODS OUTPUT InfoCrit=FitR3Day; * Save for fit; RUN;
* Test random day of the week;
%FitTest(FitFewer=FitR3QTime, FitMore=FitR3Day);
```

Does the level-2 Monday effect vary randomly over level-3 persons? *Nope, $-2\Delta LL(\sim3)$ =4.94, p = .176, so drop it*

```
Likelihood Ratio Test for FitR3QTime vs. FitR3Day
              Neg2Log
   Name        Like      Parms        AIC        BIC     DevDiff     DFdiff      Pvalue
FitR3QTime    4493.9      10       4513.9     4525.7        .           .           .
FitR3Day      4488.9      13       4514.9     4530.3     4.94274        3        0.17604
```

Btw, here is how to add an R-matrix AR(1) level-1 correlation using continuous time. It did not converge.

```
REPEATED / TYPE=SP(Pow)(SinceMidnight) SUBJECT=PersonID*Day; * Level 1 now has AR(1) corr;
```

Model 3a: Add fixed main effects of Negative Affect (NA) at each level using variable-based-centering

$$\text{Level 1 Time:}\quad \text{Fatigue}_{tdi} = \beta_{0di} + \beta_{1di}(\text{Hour}_{tdi} - 15) + \beta_{2di}(\text{Hour}_{tdi} - 15)^2 + \beta_{3di}(\text{NA}_{tdi} - \overline{\text{NA}}_{di}) + e_{tdi}$$
$$\text{Level 2 Day:}\quad \text{Intercept: } \beta_{0di} = \delta_{00i} + \delta_{01i}(\text{Monday}_{di}) + \delta_{02i}(\overline{\text{NA}}_{di} - \overline{\text{NA}}_i) + U_{0di}$$
$$\text{Linear Time: } \beta_{1di} = \delta_{10i} + U_{1di}$$
$$\text{Quadratic Time: } \beta_{2di} = \delta_{20i} + U_{2di}$$
$$\text{Within-Day NA: } \beta_{3di} = \delta_{30i}$$
$$\text{Level 3 Person: Intercept: } \delta_{00i} = \gamma_{000} + \gamma_{001}(\overline{\text{Hour}}_i - 15) + \gamma_{002}(\overline{\text{NA}}_i - 10) + V_{00i}$$
$$\text{Linear Time: } \delta_{10i} = \gamma_{100} + \gamma_{101}(\overline{\text{Hour}}_i - 15)$$
$$\text{Quadratic Time: } \delta_{20i} = \gamma_{200} + \gamma_{201}(\overline{\text{Hour}}_i - 15) + V_{20i}$$
$$\text{Within-Day NA: } \delta_{30i} = \gamma_{300}$$
$$\text{It's Monday: } \delta_{01i} = \gamma_{010}$$
$$\text{Between-Day NA: } \delta_{02i} = \gamma_{020}$$

Given their centering, **each NA effect is specific to its level**—*being grumpier than*:

L1 = the rest of the day (the day's mean)

L2 = usual (the person's mean across days)

L3 = other people (on average)

```
TITLE1 "Model 3a: Add Fixed Main Effects of Negative Affect Per Level";
```

```
PROC MIXED DATA=work.EMA COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
       CLASS PersonID Day;
       MODEL Fatigue = Monday L1T L1T*L1T L3T L1T*L3T L1T*L1T*L3T L1NA L2NA L3NA
                      / SOLUTION DDFM=Satterthwaite OUTPM=PredNA; * Save for total-R2;
       RANDOM INTERCEPT     L1T*L1T / GCORR TYPE=UN SUBJECT=PersonID;       * Level 3;
       RANDOM INTERCEPT L1T L1T*L1T / GCORR TYPE=UN SUBJECT=PersonID*Day;   * Level 2;
       ODS OUTPUT InfoCrit=FitFixNA CovParms=CovFixNA; * Save for fit and pseudo-R2; RUN;
```

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	PersonID	0.9817	0.4664	2.10	0.0177
UN(2,1)	PersonID	-0.00662	0.009453	-0.70	0.4835
UN(2,2)	PersonID	0.000459	0.000294	1.56	0.0594
UN(1,1)	PersonID*Day	1.8845	0.4082	4.62	<.0001
UN(2,1)	PersonID*Day	0.03496	0.03545	0.99	0.3240
UN(2,2)	PersonID*Day	0.01222	0.005893	2.07	0.0191
UN(3,1)	PersonID*Day	-0.01556	0.01034	-1.50	0.1325
UN(3,2)	PersonID*Day	-0.00144	0.000885	-1.63	0.1036
UN(3,3)	PersonID*Day	0.000699	0.000311	2.25	0.0124
Residual		2.9904	0.2352	12.71	<.0001

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	7.2351	0.3269	27.5	22.13	<.0001
Monday	0.4678	0.2691	243	1.74	0.0834
L1T	0.008782	0.01598	254	0.55	0.5831
L1T*L1T	0.01153	0.006086	14.3	1.89	0.0785
L3T	-0.1144	0.2599	21.6	-0.44	0.6640
L1T*L3T	-0.00765	0.01498	253	-0.51	0.6101
L1T*L1T*L3T	-0.00890	0.005504	15.6	-1.62	0.1259
L1NA	0.2283	0.03145	631	7.26	<.0001
L2NA	0.4293	0.06821	227	6.29	<.0001
L3NA	0.6205	0.08167	27.1	7.60	<.0001

Based on the change in their *p*-values, it looks like L3NA explained some of the L3 time-of-day effects.

Big picture interpretation:
Being grumpier (than the rest of the day at L1, than usual at L2, or than other people L3) is related to feeling more fatigue.

```
* Pseudo-R2 for NA main effects;
%PseudoR2(NCov=10, CovFewer=CovRandTime, CovMore=CovFixNA);
* Total-R2 for NA main effects;
%TotalR2(DV=Fatigue, PredFewer=PredTime, PredMore=PredNA);
```

Name	Pred Corr	TotalR2	R2Diff
PredNA	0.62385	0.38919	0.30684

PsuedoR2 (% Reduction) for CovRandTime vs. CovFixNA

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovRandTime	UN(1,1)	PersonID	6.5735	2.3371	2.81	0.0025	.
CovRandTime	UN(2,2)	PersonID	0.000531	0.000331	1.60	0.0546	.
CovRandTime	UN(1,1)	PersonID*Day	2.4547	0.4782	5.13	<.0001	.
CovRandTime	UN(2,2)	PersonID*Day	0.01155	0.006327	1.82	0.0340	.
CovRandTime	UN(3,3)	PersonID*Day	0.000802	0.000354	2.27	0.0117	.
CovRandTime	Residual		3.2275	0.2550	12.66	<.0001	.
CovFixNA	UN(1,1)	PersonID	0.9817	0.4664	2.10	0.0177	0.85066 for L3 int
CovFixNA	UN(2,2)	PersonID	0.000459	0.000294	1.56	0.0594	0.13539 for L3 quad ??
CovFixNA	UN(1,1)	PersonID*Day	1.8845	0.4082	4.62	<.0001	0.23229 for L2 int
CovFixNA	UN(2,2)	PersonID*Day	0.01222	0.005893	2.07	0.0191	-0.05844 for L2 linear
CovFixNA	UN(3,3)	PersonID*Day	0.000699	0.000311	2.25	0.0124	0.12821 for L2 quad ??
CovFixNA	Residual		2.9904	0.2352	12.71	<.0001	0.07347 for L1 res

It is unexpected that adding main effects of NA would reduce random quadratic time variance—in theory that should have only happened by adding interactions of NA by quadratic time. In testing those interactions with time, though, none of them was significant. So perhaps it's just a case of pseudo-R^2 being pseudo??

$$\text{Level 1 Time:}\quad \text{Fatigue}_{tdi} = \beta_{0di} + \beta_{1di}(\text{Hour}_{tdi} - 15) + \beta_{2di}(\text{Hour}_{tdi} - 15)^2 + \beta_{3di}(\text{NA}_{tdi} - \overline{\text{NA}}_{di}) + e_{tdi}$$

$$\text{Level 2 Day:}\quad \text{Intercept:}\quad \beta_{0di} = \delta_{00i} + \delta_{01i}(\text{Monday}_{di}) + \delta_{02i}(\overline{\text{NA}}_{di} - \overline{\text{NA}}_{i}) + U_{0di}$$

$$\text{Linear Time:}\quad \beta_{1di} = \delta_{10i} + U_{1di}$$

$$\text{Quadratic Time:}\quad \beta_{2di} = \delta_{20i} + U_{2di}$$

$$\text{Within-Day NA:}\quad \beta_{3di} = \delta_{30i} + U_{3di} \leftarrow \textbf{3b}$$

$$\text{Level 3 Person: Intercept:}\quad \delta_{00i} = \gamma_{000} + \gamma_{001}(\overline{\text{Hour}}_{i} - 15) + \gamma_{002}(\overline{\text{NA}}_{i} - 10) + V_{00i}$$

$$\text{Linear Time:}\quad \delta_{10i} = \gamma_{100} + \gamma_{101}(\overline{\text{Hour}}_{i} - 15)$$

$$\text{Quadratic Time:}\quad \delta_{20i} = \gamma_{200} + \gamma_{201}(\overline{\text{Hour}}_{i} - 15) + V_{20i}$$

$$\text{Within-Day NA:}\quad \delta_{30i} = \gamma_{300} + V_{30i} \leftarrow \textbf{3c}$$

$$\text{It's Monday:}\quad \delta_{01i} = \gamma_{010}$$

$$\text{Between-Day NA:}\quad \delta_{02i} = \gamma_{020} + V_{02i} \leftarrow \textbf{3d}$$

Before testing main or moderation effects of sleep, we need to test random NA effects.

Models 3b and 3c added random effects of L1 NA across L2 days and L3 persons. **Model 3d** added a random effect of L2 NA over L3 persons. Only lines of code that changed are shown.

```
TITLE1 "Model 3b: Add Random Effect of L1 NA Across L2 Days";
      RANDOM INTERCEPT L1T L1T*L1T L1NA / GCORR TYPE=UN SUBJECT=PersonID*Day; * Level 2;
      ODS OUTPUT InfoCrit=FitRand2NA1 CovParms=CovRandNA; * Save for fit, pseudo-R2; RUN;
* Test random L1 NA across days; %FitTest(FitFewer=FitFixNA, FitMore=FitRand2NA1);
```

Does the level-1 NA effect vary randomly over level-2 days? *Yes, −2ΔLL(~4) =45.92, p < .001, so keep it*

Likelihood Ratio Test for FitFixNA vs. FitRand2NA1

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitFixNA	4383.0	10	4403.0	4414.8	.	.	.
FitRand2NA1	4337.1	14	4365.1	4381.6	45.9229	4	2.5555E-9

```
TITLE1 "Model 3c: Add Random Effect of L1 NA Across L3 Persons";
      RANDOM INTERCEPT L1T*L1T L1NA / GCORR TYPE=UN SUBJECT=PersonID; * Level 3;
      ODS OUTPUT InfoCrit=FitRand3NA1; RUN;
* Test random L1 NA across persons;
%FitTest(FitFewer=FitRand2NA1, FitMore=FitRand3NA1);
```

Does the level-1 NA effect vary randomly over level-3 persons? *No, −2ΔLL(~3) =3.21, p = .361, so drop it*

Likelihood Ratio Test for FitRand2NA1 vs. FitRand3NA1

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitRand2NA1	4337.1	14	4365.1	4381.6	.	.	.
FitRand23NA1	4333.9	17	4367.9	4387.9	3.20629	3	0.36090

```
TITLE1 "Model 3d: Add Random Effect of L2 NA Across L3 Persons";
      RANDOM INTERCEPT L1T*L1T L2NA / GCORR TYPE=UN SUBJECT=PersonID; * Level 3;
      ODS OUTPUT InfoCrit=FitRand3NA2; RUN;
* Test random L1 NA across persons;
%FitTest(FitFewer=FitRand2NA1, FitMore=FitRand3NA2);
```

Does the level-2 NA effect vary randomly over level-3 persons? *No, −2ΔLL(~3) =4.97, p = .174, so drop it*

Likelihood Ratio Test for FitRand2NA1 vs. FitRand3NA2

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitRand2NA1	4337.1	14	4365.1	4381.6	.	.	.
FitRand3NA2	4332.1	17	4366.1	4386.1	4.96928	3	0.17406

Now we can introduce the effects of sleep, beginning with the main effects at each level…

Level 1 Time: $\text{Fatigue}_{tdi} = \beta_{0di} + \beta_{1di}(\text{Hour}_{tdi} - 15) + \beta_{2di}(\text{Hour}_{tdi} - 15)^2 + \beta_{3di}(\text{NA}_{tdi} - \overline{\text{NA}}_{di}) + e_{tdi}$

Level 2 Day: Intercept: $\beta_{0di} = \delta_{00i} + \delta_{01i}(\text{Monday}_{di}) + \delta_{02i}(\overline{\text{NA}}_{di} - \overline{\text{NA}}_{i}) + \delta_{03i}(\text{Sleep}_{di} - \overline{\text{Sleep}}_{i}) + U_{0di}$

Linear Time: $\beta_{1di} = \delta_{10i} + U_{1di}$

Quadratic Time: $\beta_{2di} = \delta_{20i} + U_{2di}$

Within-Day NA: $\beta_{3di} = \delta_{30i} + U_{3di}$

Level 3 Person: Intercept: $\delta_{00i} = \gamma_{000} + \gamma_{001}(\overline{\text{Hour}}_{i} - 15) + \gamma_{002}(\overline{\text{NA}}_{i} - 10) + \gamma_{003}(\overline{\text{Sleep}}_{i} - 7) + V_{00i}$

Linear Time: $\delta_{10i} = \gamma_{100} + \gamma_{101}(\overline{\text{Hour}}_{i} - 15)$

Quadratic Time: $\delta_{20i} = \gamma_{200} + \gamma_{201}(\overline{\text{Hour}}_{i} - 15) + V_{20i}$

Within-Day NA: $\delta_{30i} = \gamma_{300}$

It's Monday: $\delta_{01i} = \gamma_{010}$

Between-Day NA: $\delta_{02i} = \gamma_{020}$

Yesterday Sleep: $\delta_{03i} = \gamma_{030} + V_{03i}$ ← 4b

Model 4a added fixed main effects of L2 last night sleep (0=Person Mean) and L3 person mean sleep (0=7 hours).

Model 4b added a random effect of L2 last night sleep over L3 persons, but it did not converge.

```
TITLE1 "Model 4a: Add Fixed Main Effects of Sleep Per Level";
PROC MIXED DATA=work.EMA COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
       CLASS PersonID Day;
       MODEL Fatigue = Monday L1T L1T*L1T L3T L3T*L1T L3T*L1T*L1T L1NA L2NA L3NA
                       L2Sleep L3Sleep / SOLUTION DDFM=Satterthwaite OUTPM=PredSleep; *For total-R2;
       RANDOM INTERCEPT     L1T*L1T      / GCORR TYPE=UN SUBJECT=PersonID;       * Level 3;
       RANDOM INTERCEPT L1T L1T*L1T L1NA / GCORR TYPE=UN SUBJECT=PersonID*Day;   * Level 2;
       ODS OUTPUT InfoCrit=FitSleep CovParms=CovSleep; * Save for fit and pseudo-R2; RUN;
```

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	7.1411	0.3224	25.3	22.15	<.0001
Monday	0.5626	0.2719	238	2.07	0.0396
L1T	0.01034	0.01545	240	0.67	0.5040
L1T*L1T	0.01530	0.004980	13	3.07	0.0089
L3T	-0.1490	0.2556	19.9	-0.58	0.5667
L1T*L3T	-0.00749	0.01456	232	-0.51	0.6074
L1T*L1T*L3T	-0.00669	0.004580	14.6	-1.46	0.1652
L1NA	0.1630	0.04509	87.9	3.62	0.0005
L2NA	0.4264	0.06834	217	6.24	<.0001
L3NA	0.6330	0.08334	23.9	7.60	<.0001
L2Sleep	-0.1472	0.08092	208	-1.82	0.0703
L3Sleep	-0.2308	0.3056	30.5	-0.76	0.4559

I then examined interactions with time of day for both NA and sleep; all were nonsignificant. I also examined the intra-variable sleep interaction, and it was nonsignificant as well.

Big picture interpretation:
Getting less sleep than usual the night before is related to feeling more fatigue that day (duh). But getting less sleep than other people does *not* imply you are more tired than other people.

Name	Pred Corr	TotalR2	R2Diff
PredSleep	0.62890	0.39551	.006319902

```
* Pseudo-R2 for sleep main effects; %PseudoR2(NCov=14, CovFewer=CovRandNA, CovMore=CovSleep);
* Total-R2  for sleep main effects; %TotalR2(DV=Fatigue, PredFewer=PredNA, PredMore=PredSleep);
```

PsuedoR2 (% Reduction) for CovRandNA vs. CovFixSleep

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovRandNA	UN(1,1)	PersonID	0.8647	0.4366	1.98	0.0238	.
CovRandNA	UN(2,2)	PersonID	0.000206	0.000193	1.06	0.1435	.
CovRandNA	UN(1,1)	PersonID*Day	2.1662	0.4109	5.27	<.0001	.
CovRandNA	UN(2,2)	PersonID*Day	0.01400	0.005611	2.49	0.0063	.
CovRandNA	UN(3,3)	PersonID*Day	0.000831	0.000296	2.81	0.0025	.
CovRandNA	UN(4,4)	PersonID*Day	0.09070	0.03097	2.93	0.0017	.
CovRandNA	Residual		2.4465	0.2108	11.60	<.0001	.
CovFixSleep	UN(1,1)	PersonID	0.9004	0.4585	1.96	0.0248	-0.041346 for L3 int
CovFixSleep	UN(2,2)	PersonID	0.000212	0.000197	1.07	0.1413	-0.030228 for L3 quad
CovFixSleep	UN(1,1)	PersonID*Day	2.1559	0.4099	5.26	<.0001	0.004754 for L2 int
CovFixSleep	UN(2,2)	PersonID*Day	0.01421	0.005642	2.52	0.0059	-0.015215 for L2 linear
CovFixSleep	UN(3,3)	PersonID*Day	0.000833	0.000296	2.81	0.0025	-0.002261 for L2 quad
CovFixSleep	UN(4,4)	PersonID*Day	0.09065	0.03095	2.93	0.0017	0.000633 for L2 L1-NA
CovFixSleep	Residual		2.4451	0.2110	11.59	<.0001	0.000552 for L1 res

Level 1 Time: $\text{Fatigue}_{tdi} = \beta_{0di} + \beta_{1di}(\text{Hour}_{tdi} - 15) + \beta_{2di}(\text{Hour}_{tdi} - 15)^2 + \beta_{3di}(\text{NA}_{tdi} - \overline{\text{NA}}_{di}) + e_{tdi}$

Level 2 Day: Intercept: $\beta_{0di} = \delta_{00i} + \delta_{01i}(\text{Monday}_{di}) + \delta_{02i}(\overline{\text{NA}}_{di} - \overline{\text{NA}}_{i}) + \delta_{03i}(\text{Sleep}_{di} - \overline{\text{Sleep}}_{i})$
$+ \delta_{04i}(\overline{\text{NA}}_{di} - \overline{\text{NA}}_{i})(\text{Sleep}_{di} - \overline{\text{Sleep}}_{i}) + U_{0di}$

Linear Time: $\beta_{1di} = \delta_{10i} + U_{1di}$

Quadratic Time: $\beta_{2di} = \delta_{20i} + U_{2di}$

Within-Day NA: $\beta_{3di} = \delta_{30i} + \delta_{33i}(\text{Sleep}_{di} - \overline{\text{Sleep}}_{i}) + U_{3di}$

Level 3 Person: Intercept: $\delta_{00i} = \gamma_{000} + \gamma_{001}(\overline{\text{Hour}}_{i} - 15) + \gamma_{002}(\overline{\text{NA}}_{i} - 10) + \gamma_{003}(\overline{\text{Sleep}}_{i} - 7)$
$+ \gamma_{004}(\overline{\text{NA}}_{i} - 10)(\overline{\text{Sleep}}_{i} - 7) + V_{00i}$

Linear Time: $\delta_{10i} = \gamma_{100} + \gamma_{101}(\overline{\text{Hour}}_{i} - 15)$

Quadratic Time: $\delta_{20i} = \gamma_{200} + \gamma_{201}(\overline{\text{Hour}}_{i} - 15) + V_{20i}$

Within-Day NA: $\delta_{30i} = \gamma_{300} + \gamma_{303}(\overline{\text{Sleep}}_{i} - 7)$

It's Monday: $\delta_{01i} = \gamma_{010}$

Between-Day NA: $\delta_{02i} = \gamma_{020} + \gamma_{023}(\overline{\text{Sleep}}_{i} - 7)$

Yesterday Sleep: $\delta_{03i} = \gamma_{030} + \gamma_{002}(\overline{\text{NA}}_{i} - 10)$

Between-Day NA by Yesterday Sleep: $\delta_{04i} = \gamma_{040}$

Within-Day NA by Yesterday Sleep: $\delta_{33i} = \gamma_{330}$

Model 4c added six total interactions of L2 and L3 sleep with each level of NA.

This will be our final model to report!

```
TITLE1 "Model 4c: Add Fixed Effects of Sleep Moderating Effects of NA";
PROC MIXED DATA=work.EMA COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
      CLASS PersonID Day;
      MODEL Fatigue = Monday L1T L1T*L1T L3T L3T*L1T L3T*L1T*L1T L1NA L2NA L3NA
                      L2Sleep L2Sleep*L1NA L2Sleep*L2NA L2Sleep*L3NA
                      L3Sleep L3Sleep*L1NA L3Sleep*L2NA L3Sleep*L3NA
                          / SOLUTION DDFM=Satterthwaite OUTPM=PredModSleep; * Save for total-R2;
      RANDOM INTERCEPT      L1T*L1T      / GCORR TYPE=UN SUBJECT=PersonID;       * Level 3;
      RANDOM INTERCEPT L1T L1T*L1T L1NA / GCORR TYPE=UN SUBJECT=PersonID*Day;    * Level 2;
      ODS OUTPUT CovParms=CovModSleep; * Save for fit and pseudo-R2; RUN;
```

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	7.0175	0.3835	22.2	18.30	<.0001
Monday	0.5671	0.2711	234	2.09	0.0375
L1T	0.009415	0.01546	241	0.61	0.5430
L1T*L1T	0.01578	0.004903	13	3.22	0.0067
L3T	-0.1103	0.2787	20.1	-0.40	0.6964
L1T*L3T	-0.00875	0.01457	235	-0.60	0.5491
L1T*L1T*L3T	-0.00684	0.004511	14.6	-1.52	0.1508
L1NA	0.1273	0.04610	83.6	2.76	0.0071
L2NA	0.3887	0.07120	215	5.46	<.0001
L3NA	0.5991	0.09694	21.4	6.18	<.0001
L2Sleep	-0.1346	0.1226	197	-1.10	0.2733
L1NA*L2Sleep	-0.02965	0.03994	74	-0.74	0.4602
L2NA*L2Sleep	0.02243	0.05975	211	0.38	0.7077
L3NA*L2Sleep	0.004104	0.03121	184	0.13	0.8955
L3Sleep	0.6632	0.5490	61	1.21	0.2317
L1NA*L3Sleep	-0.1950	0.07165	97.8	-2.72	0.0077
L2NA*L3Sleep	-0.2110	0.09964	240	-2.12	0.0352
L3NA*L3Sleep	0.2637	0.1453	60.4	1.82	0.0745

Big picture interpretation:
People who sleep more than others have *weaker* within-person effects of negative affect on fatigue—they appear *less* susceptible to feeling more tired when they are grumpy (or vice-versa). But people who sleep more have (almost) a *greater* between-person effect of negative affect—the tendency for grumpy people to be tired people is (almost) stronger in people who sleep more.

There are no level-2 moderating effects of having slept less the night before, however.

```
* Pseudo-R2 for sleep moderating effects;
%PseudoR2(NCov=14, CovFewer=CovSleep, CovMore=CovModSleep);
* Total-R2 for sleep main effects;
%TotalR2(DV=Fatigue, PredFewer=PredSleep, PredMore=PredModSleep);
```

Name	Pred Corr	TotalR2	R2Diff
PredModSleep	0.63195	0.39936	.003844363

PsuedoR2 (% Reduction) for CovFixSleep vs. CovModSleep

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovFixSleep	UN(1,1)	PersonID	0.9004	0.4585	1.96	0.0248	.
CovFixSleep	UN(2,2)	PersonID	0.000212	0.000197	1.07	0.1413	.
CovFixSleep	UN(1,1)	PersonID*Day	2.1559	0.4099	5.26	<.0001	.
CovFixSleep	UN(2,2)	PersonID*Day	0.01421	0.005642	2.52	0.0059	.
CovFixSleep	UN(3,3)	PersonID*Day	0.000833	0.000296	2.81	0.0025	.
CovFixSleep	UN(4,4)	PersonID*Day	0.09065	0.03095	2.93	0.0017	.
CovFixSleep	Residual		2.4451	0.2110	11.59	<.0001	.
CovModSleep	UN(1,1)	PersonID	1.0858	0.5173	2.10	0.0179	-0.20588 for L3 int
CovModSleep	UN(2,2)	PersonID	0.000199	0.000191	1.04	0.1485	0.06259 for L2 quad
CovModSleep	UN(1,1)	PersonID*Day	2.0173	0.3979	5.07	<.0001	0.06429 for L2 int
CovModSleep	UN(2,2)	PersonID*Day	0.01436	0.005630	2.55	0.0054	-0.01062 for L2 linear
CovModSleep	UN(3,3)	PersonID*Day	0.000813	0.000294	2.77	0.0028	0.02409 for L2 quad
CovModSleep	UN(4,4)	PersonID*Day	0.08324	0.02960	2.81	0.0025	0.08167 for L2 L1-NA
CovModSleep	Residual		2.4456	0.2104	11.62	<.0001	-0.00019 for L1 res

Sample Results Section:

The extent to which daily sleep moderates the association between self-reported negative affect and fatigue was examined in a sample of 25 adolescents who completed surveys about their mood and energy levels up to four times per day for up to 20 days (total N = 976). All analyses were conducted using multilevel models estimated with residual maximum likelihood in SAS MIXED. Accordingly, the significance of fixed effects was evaluated with Wald tests using Satterthwaite denominator degrees of freedom, whereas the significance of random effects was evaluated via likelihood ratio tests (i.e., −2ΔLL with degrees of freedom equal to the number of new random effects variances and covariances). All results for the likelihood ratio tests referenced below are shown in Table X. Effect sizes are reported using two metrics: pseudo-R^2, the proportion reduction in each variance component after including the fixed effects, and total-R^2, the square of the correlation between the original fatigue outcome and the outcome predicted by the model's fixed effects.

Given the intensive longitudinal sampling design, we first examined the extent of dependency of observations from the same person, as well as from the same day, by contrasting the fit of two empty means models (i.e., with no predictors). Specifically, we compared the fit of two-level models of time within persons, in which only persons had a random intercept, against the fit of three-level models, in which a random intercept for day was added. As reported in Table X, a three-level model fit significantly better for both fatigue and negative affect. For fatigue, 52% of the total variance was between persons (level 3), 13% was between days (level 2), and 36% was within days (level 1); for negative affect, these values were 77%, 12%, and 21%, respectively. Given that participants could select at what time they completed each of the four surveys during the day, we also examined the same two- and three-level models treating time of day as an outcome. While there was no significant variance across days (as expected given that the survey timing was intended to be constant across days), 2.4% of the total variance in time of day was between persons. Finally, we examined a two-level model treating last night's sleep as an outcome, which revealed that 27% of the variance in sleep was between persons.

We then explored differences by time of day at level 1 by rounding time to the nearest half hour and estimating a saturated means model (i.e., with a separate mean for each time value). Fatigue followed a U-shape across the day, peaking in the mornings and in the late evening. We then approximated this pattern by fitting linear and quadratic level-1 effects of time of day, as well as their interactions with (and a main effect of) a level-3 predictor for person mean time of day; all time variables were entered such that 0 = 3 PM. The level-3 person mean time effects were necessary to create contextual effects by which the level-1 time of day effects then represented purely within-day effects as desired. We also examined potential level-2 time effects by estimating saturated means models for day in study and for day of the week. While no discernable pattern of change in fatigue by day of study was observed, fatigue appeared to be greatest on Mondays, which we approximated with a single contrast in which 1 = Monday and 0 = other days. Together, these fixed effects related to time accounted for 8.23% of the total variance in fatigue (i.e., total-R^2 = .083), including 4.57% of the level-3 person random intercept variance, 3.60% of the level-2 day random intercept variance, and 3.10% of the level-1 within-day residual variance.

We next examined the potential for random effects of these time-related predictors, beginning with level-1time of day. Given the heavily quadratic pattern, a random quadratic effect of time of day over level-2 days was fitted first, followed by a random linear effect. Each resulted in significant model improvement, indicating that the U-shaped trend for fatigue varied randomly across days within persons. Retaining these day-level random effects, we then examined random quadratic and linear time of day effects over level-3 persons. Only the random quadratic time of day effect varied significantly over persons, and thus the random linear time of day effect at level 3 was not retained. A random effect over level-3 persons for the difference in fatigue on Mondays did not improve model fit and was also not retained.

We then added level-specific predictors of negative affect using variable-based-centering for levels 1 and 2. Specifically, the level-1 within-day predictor was centered at the day's mean, the level-2 day mean predictor was centered at the person's mean, and the level-3 person mean predictor was centered at 10 (near the sample mean). All three effects were significantly positive and together accounted for an additional 30.7% of the total variance. The level-1 effect of being grumpier than the rest of the day explained 7.35% of the remaining level-1 residual variance. The level-2 effect of having a grumpier day than usual explained 23.2% of the remaining level-2 day random intercept variance, as well as—inexplicably—12.8% of the level-2 random quadratic time of day variance. Similarly, the level-3 effect of being grumpier than other people explained 85.1% of the remaining level-3 person random intercept variance, as well as 13.5% of the level-3 random quadratic time of day variance. In then testing random effects, we found a significant random effect of level-1 negative affect across level-2 days, which was retained in the model. There was no significant random variance across level-3 persons in the level-1 or level-2 effects of negative affect, and so these random effects were not retained.

We next added fixed main effects of level-2 previous night's sleep (centered at the person's mean) and level-3 person mean sleep (centered at seven hours, near the sample mean). Both effects were nonsignificantly negative and accounted for an additional 0.63% of the total variance. The level-2 effect of having less sleep than usual the night before explained 0.48% of the remaining level-2 day random intercept variance, and the level-3 effect of averaging less sleep than other people did not reduce the remaining level-3 person random intercept variance. A model also testing the random variance across level-3 persons in the level-2 sleep effect did not converge. All interactions of negative affect and sleep with time of day were nonsignificant; an intra-variable interaction of level-2 by level-3 sleep was also nonsignificant.

Finally, to examine the moderating effects of sleep on the relationship between negative affect and fatigue, we added the six possible two-way interactions between negative affect and sleep at each level, which accounted for an additional 0.38% of the variance, making the model total-R^2 = .399. The remaining level-2 day random intercept variance was reduced by 6.43% from three interactions: level-2 sleep by level-2 negative affect, level-2 sleep by level-3 negative affect, and level-3 sleep by level-2 negative affect. The remaining level-2 day random variance in the effect of level-1 negative affect was reduced by 8.17% from two interactions: level-2 sleep by level-1 negative affect, and level-3 sleep by level-1 negative affect. The remaining level-3 person random intercept variance was not reduced by the level-3 sleep by level-3 negative affect interaction. Inexplicably, the remaining level-2 and level-3 random quadratic time of day variances were also reduced by 2.41% and 6.26%, respectively.

Results for this final model are shown in Table XX and can be interpreted as follows. Fatigue was significantly higher on Mondays. There remained a significant positive quadratic trend of level-1 time of day (as evaluated for a person with a mean time of day = 3 PM), although no contextual effects of level-3 person mean time of day were significant. The simple main effects of negative affect at each level (each conditional on the reference values for level-2 and level-3 sleep) remained significantly positive, indicating greater negative affect was related to greater fatigue. The simple main effects of level-2 and level-3 sleep (each conditional on the reference values for level-1, level-2, and level-3 negative affect) remained nonsignificant. The three interactions of level-2 sleep with each of level of negative affect were all nonsignificant, indicating no differentiable impact of getting less sleep than usual on the association between negative affect and fatigue. A different pattern of results was found with respect to level-3 sleep: the interactions of level-3 person mean sleep with level-1 and level-2 negative affect were both significantly negative, indicating that people who sleep more than others have weaker within-person effects of negative affect on fatigue—they appear less susceptible to feeling more tired than usual when they are grumpier than usual (or vice-versa). In contrast, the interaction of level-3 person mean sleep with level-3 negative affect was almost significantly positive (p = .075), indicating that the tendency for grumpy people to be tired people is (almost) stronger in people who sleep more than other people.